OPMT 5701

Lab 1: The Basics of MAPLE

NUMBERS

The arithmetic operations are represented in Maple by +, -, *, / and ^.

```
> a:=(3+4)/5;
> b:=(3*4)/5;
> c:=exp(1)^2;
```

For precision of calculation, Maple does not convert fractions or expressions with exponents unless it is forced to by making one of the numbers decimal such as replacing 3 by 3. or by using the evalf command. exp(1) is Maple's way of inputting the number e.

```
> (3.+4)/5;
> evalf(a);
> evalf(exp(1));
> evalf(Pi);
> 3^(1/3);
> evalf(3^(1/3));
```

You can use Maple to check your arithmetical skills. For example use +, * and brackets to write an expression involving 3, 4 and 7 which gives
the answer 49. Does (3/4)+(5/6) = (3+5)/(4+6)? Check it out. What is an alternative expression for calculating (3/4)*(5/6)?

Maple does symbolic calculations as well as numerical calculations.

```
> (x+y)/(w+z);
> (x/w)+(y/z);
> simplify(%);
>
```

The answer is not zero so the two expressions were not equal. You can also use Maple to review the laws of exponents.

```
> (x^m)+(x^n);
> simplify(%);
>
> (x^m)*(x^n);
> simplify(%);
> (x^m)^n;
> simplify(%);
>
```

Maple doesn't recognize the second law of exponents amd try the following examples to show that Maple can make mistakes.

```
> (5^(3/2))^(3/4);
> 5^((3/2)*(3/4));
> evalf(%);
> 5^(9/8);
```

FUNCTIONS

Maple has lots of built-in functions for you to use. Examples of getting values of these functions are:

```
> restart:
> exp(1);

> ln(1);
> log[10](10);
> sqrt(Pi);
```

Calculate some other values of these functions by clicking the mouse on the number in the expressions above, deleting the number, entering your own number and pressing ENTER

.

The rules of logaritms can be reviewed.. The assumptions are necessary to make Maple stick to real numbers. The ~ is telling you that some assumption has been made about the variable.

```
> assume(x>0);
```

```
> assume(y,real);
> ln(x+y);
> simplify(%);
> ln(x*y);
> simplify(%);
> ln(x^y);
> simplify(%);
> ln(exp(y));
> exp(ln(x));
```

You can also plot these functions.

```
> plot(ln(x),x=.5..2);
> plot(exp(y),y=-1..3);
> plot(cos(y),y=0..4*Pi);
```

You can also define your own functions in Maple. The next exercise in this assignment is to define and plot an average cost curve.

```
> assume(x>=0);
> f:=x->x^2-20*x+120;
> f(0);
> f(5);
```

Calculating f(10), f(15) and f(20) as well should give you an idea of what the plot is going to look like.

```
> plot(f(x),x=0..20);
```

As economists, we probably don't like the way Maple does this plot. The scale for the vertical axis can be changed to 0 to 150 (above f(0)). Note that if you click on a plot the plot will be put in a box and the menus at the top of the screen will change to give you some options in the way the plot is drawn. Check out the style and axes menus. If you click on a point in the plot window the coordinates of the point will be shown in the toolbar at the left.

```
> plot(f(x),x=0..20,y=0..150);
```

More accuracy can be obtained in getting the values of the minimum average cost point (by clicking on it) by reducing the range for x and y in the last command.

LEVEL CURVES

The commands here illustrate the plotting of level curves , e.g. contour lines. Again if you click the plot you get some menus at the top which can be used to change the appearance of the plot,

```
> with(plots);
> contourplot(x^.25+x*y+y^.25,x=0..5,y=0..5);
> implicitplot({x^.25+x*y+y^.25=2,x^.25+x*y+y^.25=4,x^.25+x*y+y^.25=6},x
> =0..5,y=0..5);
```

Using implicitplot gives more control over the levels plotted. You can change the levels to be plotted by editing and reexecuting the above command. You can also add more levels if you wish.

CURVATURE OF SURFACES

```
> U:=(x,y)->x^(1/4)*y^(1/2);
> plot3d(U(x,y),x=0..5,y=0..10-2*x,orientation=[-15,45]);
```

This surface is concave. Click the plot and look at the effects of changing the axes or the style of the plot. When you make a change, the plot disappears into a white box. Click the REDRAW button marked R at the end of the third toolbar to see the recalculated plot. The orientation of the axes can be changed on the second toolbar at the left or by pointing at the white box and dragging the pointer.
Now go back and change (1/4,1/2) to (1,3/4) or to anything where the sum of the 2 powers is > 1. The resulting surface is neither concave or convex. It is however quasiconcave. You can check this by using the contourplot or implicitplot.commands on U(x,y).

If you wish to print any of your plots, execute the command below, then go back and reexecute your plot. The plot will appear in a window and can be printed by choosing PRINT in the FILE menu. Choosing CLOSE on the FILE menu will get you back to your worksheet.

```
> plotsetup(window);
```

```
> plot3d(U(x,y),x=0..5,y=0..10-2*x,orientation=[-15,45]);
>
```

```
>
>
>
```

OPMT 5701

Lab 2: Derivative with MAPLE

Introduction to Derivatives

Derivatives can be done in Maple either symbolically or with the answer given as a function

```
> restart:
```

```
> diff(x^n,x);
```

If you do not like the way the answer looks you can try

```
> simplify(%);
```

Here is a simple quadratic

```
> y:=10*x-x^2;
```

Take the derivative

```
> dy:=diff(y,x);
```

Now try the same type of function except using more generic parameters

```
> z:=a*x-b*x^2;
```

```
> dz:=diff(z,x);
```

Try differentiating something of your own choice.

```
>
```

RULES OF DIFFERENTIATION

The next commands illustrate the product rule, the quotient rule and give you a couple of opportunities to check that you can use the chain rule correctly.

```
> diff(x^3*ln(x+1),x);
```

```
> diff(x^3,x)*ln(x+1)+x^3*diff(ln(x+1),x);
```

```
> diff(x^3/ln(x+1),x);
```

```
> (diff(x^3,x)*ln(x+1)-x^3*diff(ln(x+1),x))/ln(x+1)^2;
```

```
> simplify(%);
```

Now using the cost function from the previous assignment, the relation between average and marginal cost will be illustrated.

```
> C:=x->x^3-4*x^2+6*x;
```

```
> diff(C(x),x);
```

```
> MC:=unapply(%,x);
```

```
> C(x)/x;
```

```
> simplify(%);
> AC:=unapply(%,x);
> plotsetup(inline);
> with(plots):
> plot(MC(x),x=0..3);
> plot(AC(x),x=0..3);
> plot([MC(x),AC(x)],x=0..3);
```

Did you click on the plot to get the coordinates of the minimum AC point?

The range for y is specified above as otherwise Maple would put the bottom of the y-range very close to the minimum MC value.

The next commands show you two ways to get MAC the derivative of AC.

```
> diff(AC(x),x);
> MAC:=unapply(%,x);
> (MC(x)-AC(x))/x;
> simplify(%);
```

You can see that you get the same answer in both cases.
You can also see at what value of x AC is at its minimum. You can also find this point numerically, The command fsolve gives only the real solutions of the equation.

```
> fsolve(MAC(x)=0,x);
> [AC(%),MC(%)];
> solve(AC(x)=MC(x),x);
>
```

Go back and add a fixed cost term to C and AC (not too large) and reexecute the commands above. You may want to change the upper bounds for x and y to get a better looking plot. It may be necessary to change the last command to fsolve to eliminate complex roots.

PARTIAL DERIVATIVES

```
> restart:
> z:=a*x+b*y;
> diff(z,x);
```

```
> diff(z,y);
> u:=x^a*y^b;
> diff(u,x);
> simplify(%);
> diff(u,y);
> simplify(%);
```

Combining derivative commands: Finding the Marginal Rate of Substitution

```
> mrs:=diff(u,x)/diff(u,y);
```

OPMT 5701

Lab 3: Matrix Algebra in MAPLE

The first command loads the matrix algebra operations into Maple.
Change the : to ; if you wish to see all the commands in this package.

```
> with(linalg);
```

```
> ?linalg
```

Next define a few matrices. You can change the numbers to any numbers you like.

```
> M1:=matrix(3,3,[1,0,-1,1,5,7,3,3,9]);
```

```
> M2:=matrix(3,3,[1,0,-1,3,6,21,3,3,9]);
```

The next commands create some other matrices from M1 and M2.

```
> M3:=submatrix(M2,2..3,1..2);
```

```
> M4:=submatrix(M1,2..3,1..3);
```

```
> M5:=stackmatrix(M2,M4);
```

```
> M6:=concat(M3,M4);
```

The next command illustrates multiplication by a scalar. Evalm is needed for the answer to appear
on the screen. Refer back to M6 to see what has happened

```
> evalm(3*M6);
```

The aritmetic operations for matrices are + , - , &* , ^ . For easier reference the next command
brings M1 and M2 back on to the screen side by side in a 3X6 matrix. Execute the next
command and after you are sure you understand what has happened execute the command agaiin
with + changed first to - and then to &* .

```
> concat(M1,M2);
```

```
> evalm(M1+M2);
```

If you wish to check your understanding of matrix multiplication the following command calculates the
[2,2] entry of the product matrix. Change the numbers in this command

to calculate some other

entries in the product matrix. If one of the numbers is negative put

it in brackets.

```
> (1*0)+(5*6)+(7*3);
```

The next comand does a power of a matrix. You can check the answers

out in the same way the

product was checked.

```
> evalm(M1^2);
```

The following illustrate some of the properties of matrix algebra. A

matrix is transposed when its

rows and columns are switched.. The properties are the associative ,

commutative and distributive

properties.

```
> M7:=evalm(transpose(M4)&*M4);
> evalm(M4);
> evalm(M7);
> ?evalm
> evalm((M1+M2)+M7);
> evalm(M1+(M2+M7));
> evalm((M1&*M2)&*M7);
> evalm(M1&*(M2&*M7));
> evalm((M1+M2)-(M2+M1));
```

The commutative law does not however work with multiplication.

```
> evalm((M1&*M2)-(M2&*M1));
```

The final property illustrated here is one of the distributive laws.

```
> evalm((M1+M2)&*M7);
> evalm((M1&*M7)+(M2&*M7));
```

Diagonal matrices are created with the diag command. This can be used

to create an identity matrix.

```
> diag(-3,2,Pi);
> iden:=diag(1,1,1);
```

Inverse matrices are either A^(-1) or inverse(A). The following

calculate the inverse, check that it

works and illustrate the properties of inverses. You should also try
to calculate the inverse of M2 .

```
> M8:=M1^(-1);
> evalm(%);
> evalm(M1&*M8-iden);
> evalm(M8&*M1-iden);
> evalm(M8^(-1));
> M9:=evalm(M1+M7);
```

The next commands illustrate the correct and the incorrect way to get
the inverse of the product of twp matrices.

```
> evalm((M1&*M9)^(-1));
> evalm((M9^(-1))&*(M1^(-1)));
> evalm((M1^(-1))*(M9^(-1)));
> evalm((M1^(-1))+(M9^(-1)));
> evalm((M1+M9)^(-1));
>
```

The last two commands have illustrated a dumb artihmetical mistake you
should not make even with numbers.

```
> restart:
> with(linalg):
```

The following shows how Maple calculates a determinant. You can of course
put any numbers that you like into M

```
> M:=matrix(3,3,[1,0,-1,3,6,2,3,-3,1]);
> det(M);
>
```

Diagonal matrices are created with the diag command. This can be used
to create an identity matrix.

```
> iden:=diag(1,1,1);
```

Inverse matrices are either M^(-1) or inverse(M). The following
calculate the inverse, check that it
works and illustrate the properties of inverses.

```
> M1:=M^(-1);
> evalm(%);
```

```
> evalm(M&*M1-iden);
> evalm(M1&*M-iden);
> evalm(M1^(-1));
```

To check that you understand how to calculate the inverse matrix by hand replace the ? in the command below by appropriate entries from M

```
> evalm(M);
> M1??:=((-1)^?)*det(matrix(2,2,[?,?,?,?]))/det(M);
>
> evalm(M1);
```

PROPERTIES OF DETERMINANTS

The commands below check some of the properties of determinants. You should look to see that the new matrices being created are what you expect them to be.

```
> evalm(M);
> M3:=transpose(M);
> det(M3);
> M4:=swaprow(M,1,3);
> det(M4);
> M5:=swapcol(M4,1,2);
> det(M5);
> M6:=mulrow(M5,3,0.3);
> det(M6);
> det(M5)*0.3;
> M7:=mulcol(M6,2,3);
> det(M7);
> det(M6)*3;
> M8:=addrow(M7,1,3,.5);
> det(M8);
> M9:=addcol(M8,3,2,1.5);
> det(M9);
>
```

M1 was the inverse matrix of M

```
> det(M);
```

```
> det(M1);
```

The determinant of the product is the product of the determinants.

This does not work with sums however.

```
> det(M&*M9);
```

```
> det(M)*det(M9);
```

```
> det(M+M9);
```

```
> det(M)+det(M9);
```

```
>
```